



Title	PC8801を使用した1200BPS端末
Author(s)	藤井, 博
Citation	大阪大学大型計算機センターニュース. 1982, 47, p. 81-92
Version Type	VoR
URL	https://hdl.handle.net/11094/65549
rights	
Note	

The University of Osaka Institutional Knowledge Archive : OUKA

<https://ir.library.osaka-u.ac.jp/>

The University of Osaka

PC8801を使用した1200BPS端末

大阪大学大型計算機センター

共同利用掛 藤 井 博

1. は し め に

1200 BPS の音響カプラーが、市販されていらい手軽に 1200 BPS で端末を接続できるようになった。

しかし、パーソナル・コンピュータを BASIC でインテリジェント端末として使用している場合、多くは 300 BPS が限度の様であり、1200 BPS での端末として活用出来る可能性は、残念ながら少ないと思われていた。

ところが、前号のセンターニュース Vol. 12, No. 2 で紹介したプログラムのロジックを一部修正するだけで、BASIC でも 1200 BPS の回線スピードに追従することが判明したので、一部未完成の所もあるが報告することにした。

2. 1200 BPS への対応方法

1200 BPS への対応方法としては、BASIC の使用をやめ機械語でプログラムを作成する方法と、BASIC プログラムでできるだけ不用な所をとりのぞき、受信バッファからの 1 文字入力をやめ、一括入力をする方法があり、今回は後者を採用した。

受信バッファからの、一括入力の方法は「LINE INPUT #1, R\$」を使用する方法と、「INPUT\$(LOC(1), #1)」を使用する方法がある。前者の場合注意することは、一旦 LINE INPUT 文の実行に入ってしまうと、その後 CR コード(16 進 0D)を受信しないかぎり LINE INPUT 文の実行から抜け出せない事態に陥ることになる。ACOS の場合附録で説明することにより LINE INPUT 文の使用を見合せた。

また、受信バッファから一括入力するため、1 文字ずつ検査して DEL コード(16 進 7F)を削除することはやめ、機械語のサブルーチンにて DEL コード(16 進 7F)を NULL コード(16 進 00)に変更することにした。

蛇足ではあるが、この DEL コード(16 進 7F)は、パソコン等でターミナル・プログラムを作成する時、非常に厄介なものであり出来れば無い方がよいのである。ACOS に接続する時、オプションによって DEL コード(16 進 7F)を付加しない様には出来ないものだろうか。

3. プログラムの説明

プログラムの大部分は、センタースタッフVol.12, No.2 で紹介したプログラムなので修正部分についてのみ説明する。

```

10 'Terminal program for NEC PC8801 by H.FUJII.
20 'Initial Set
21 CLEAR :&HFFFF:DEFINT A-Z:CONV=&HE000
22 'Machine Language Program Load
23 FOR I=&HE000 TO &HE01F
24 READ DA$:DA$=VAL("&H"+DA$):POKE I,DA
25 NEXT I
26 DATA 56,23,7E,23,66,6F,14,15,C8,7E,EE,7F,CA,13,E0,23
27 DATA C3,07,E0,36,00,23,C3,07,E0,00,00,00,00,00,00
30 WIDTH 80,25:CONSOLE .0,25,0,1:COLOR 4:PRINT CHR$(&HC):'Display set
40 CR$=CHR$(&HD):LOGING=0:DEL$=CHR$(&H7F):LF$=CHR$(&HA):GS=0:FHR=0:LSW=0
50 LISTING=0:GDP=0:GS$=CHR$(&H1D):ESC$=CHR$(&H1B):CAN$=CHR$(&H1B):FF$=CHR$(&HC):
ETB$=CHR$(&H17):E0$=CHR$(&HE0):H20$=CHR$(&H20):H30$=CHR$(&H30):H40$=CHR$(&H40):U
S$=CHR$(&H1F):H16$=CHR$(&H16)
51 HID$=CHR$(&H1D):H8$=CHR$(&H8):H1C$=CHR$(&H1C):H1E$=CHR$(&H1E)
60 'RS232C. open.
70 OPEN="com:E71.S".AS #1:'RS232C. initial set.
80 'Function key set
90 HELP ON:KEY ON:STOP ON:'Activate all function keys and help key
100 ON HELP GOSUB 890:ON STOP GOSUB 650:ON ERROR GOTO 400
110 ON KEY GOSUB 490,570,610,330,950,680,730,760,920,530

```

行番号

10 コメント。

20 コメント。

21 機械語をロードするため、BASICの使用メモリーの上限定義、AからZで始まる変数の整数型定義と、機械語サブルーチンのエントリアドレス定義。

22 コメント。

23

24 } 機械語のロード。

25

26

27 } 機械語サブルーチン。

30 ディスプレイを1行80文字25行カラーモードにセットし画面をクリア。

40

50 } 各変数を定義。

51

60

60 } Vol.12, No.2 参照。

110

```

120 'Main program
130 BL=LOG(↓):IF BL=0 THEN 220
140 'Receive routine
150 R$=INPUT$(BL, #1)
160 CALL CONV(R$)
170 PRINT R$;
180 IF .LOGING THEN PRINT #2,R$;
190 IF LISTING THEN LPRINT R$;
200 GOTO 130
210 'Send routine
220 K$=INKEY$:IF K$="" THEN 130
230 IF K$=HID$ THEN GOSUB 300:GOTO 130
240 IF K$=H8$ THEN GOSUB 300:GOTO 130
250 IF K$=H1C$ THEN 130
260 IF K$=H1E$ THEN 130
270 IF K$=US$ THEN 130
271 IF K$=ESC$ THEN 1000
280 COLOR 6:PRINT K$;:COLOR 4:PRINT...#1,K$;
290 IF LISTING THEN LPRINT K$;
291 GOTO 130

```

行番号

- 120 コメント。
- 130 ホストからの受信チェック。
- 140 コメント(メインループ内のコメントなので、出来れば無い方がよい。)
- 150 受信バッファからの一括入力。
- 160 機械語サブルーチンをコール。DELコード(16進7F)をNULLコード(16進00)に変換する。
- 170 受信データの画面表示。
- 180 フロッピーディスク書き込み指定があれば、フロッピーディスク出力。
- 190 ハードコピー指定があれば、プリント出力。(この行は、300 BPS との互換性のためのものであり、1200 BPSのみで使用される場合は、プリンターの印字速度が追いつかないので無い方がよい。)
- 200 受信チェックへ戻る。
- 210 コメント。
- 220 キーボード入力のチェック。
- 230 ← キー入力なら、バックスペース処理へ。
- 240 CTRL + H キー入力なら、バックスペース処理へ。
- 250 → キー入力を無視する。
- 260 ↑ キー入力を無視する。
- 270 ↓ キー入力を無視する。
- 271 ESC キー入力なら、ファイルハンドラー処理ルーチンへ。
- 280 キーボード入力文字の画面表示と、ホストへの送付。
- 290 ハードコピー指定があれば、プリント出力。
- 291 受信チェックへ戻る。

```

300 'Back space routine
310 PRINT #1,CHR$(8);:LOCATE POS(0)-1,CSRLIN.
320 PRINT " ";:LOCATE POS(0)-1,CSRLIN:RETURN
330 'edit mode routine.
340 Y=CSRLIN:X=POS(0)
350 COLOR 6:LINE INPUT K$:COLOR 4.
360 PRINT #1,K$;CR$;
370 LOCATE X,Y
380 RETURN
390 'Error routine
400 COLOR 2.
410 IF ERR=23 THEN PRINT "Line buffer overflow":GOTO 470
420 IF ERR=62 OR ERR=56 THEN PRINT "Bad file name":GOTO 470
430 IF ERR=60 THEN PRINT "Disk space full":GOTO 470
440 IF ERR=53 THEN PRINT "File not found":GOTO 470
450 IF ERR=5 THEN PRINT "Back Space over":GOTO 470
460 ON ERROR GOTO 0
470 COLOR 4:RESUME 480
480 RETURN
490 'log-on routine
500 PRINT #1,"$$$CON,TSS",CR$;
510 PRINT "$$$CON,TSS"
520 RETURN
530 'terminal mode end routine
540 HELP OFF:KEY OFF:STOP OFF:CLOSE #1,#2
550 ON ERROR GOTO 0
560 END
570 'send user-id and password routine.
580 PRINT #1,"KADAIBANGO1$PASSWORD1";CR$;
590 PRINT "KADAIBANGO1"
600 RETURN.
610 PRINT #1,"KADAIBANGO2$PASSWORD2";CR$;
620 PRINT "KADAIBANGO2"
630 RETURN.
640 'send break routine
650 COLOR 2:PRINT "break";:COLOR 4
651 IF FHR THEN LSW=1:RETURN
660 FOR W=1 TO 100:OUT &H21,&H3F:NEXT W:OUT &H21,&H3F
670 RETURN
680 'write file open routine
690 COLOR 2:INPUT "Enter write file name";F$:COLOR 4
700 IF F$="FILES" OR F$="files". THEN FILES:GOTO 690
710 OPEN F$ FOR OUTPUT AS #2:LOGING=1
720 RETURN
730 'Stop the write file routine
740 COLOR 2:PRINT F$;" File close":COLOR 4
750 CLOSE #2:LOGING=0:RETURN

```

行番号

300 }
) } Vol.12, No.2 参照。
 650 }

651 ファイルハンドラー処理ルーチンを実行中はLSWを1にする。

660 }
) } Vol.12, No.2 参照。
 750 }

```

760 'Read file open routine
770 COLOR 2:INPUT "Enter read file name";F$:COLOR 4
780 IF F$="FILES" OR F$="files" THEN FILES:GOTO 770
790 OPEN F$ FOR INPUT AS #2
800 LINE INPUT #2,L$
810 IF EOF(2) THEN 870
820 IF LOGING THEN 870
830 PRINT L$:PRINT #1,L$;LF$
850 IF LISTING THEN LPRINT L$
860 GOTO 800
870 PRINT L$:PRINT #1,L$;LF$:PRINT "$$$EOF":PRINT #1,"$$$$EOF";CR$;:CLOSE #2:LOGI
NG=0:RETURN
880 'Help key routine
890 PRINT "f.1=logon      f.2=userid 1   f.3=userid 2 f.4=edit      f.5=list"
900 PRINT "f.6=file(w)    f.7=write-stop f.8=file(r)   f.9=read-stop  f.10=end"
910 RETURN
920 'Stop the read file routine
930 COLOR 2:PRINT F$;" File close":COLOR 4
940 LOGING=1:RETURN
950 'Listing key on routine
960 IF LISTING THEN LISTING=0:OUT &H10,&H13:OUT &H40,0:OUT &H40,1 ELSE LISTING=1
:OUT &H10,&H11:OUT &H40,0:OUT &H40,1:LPRINT CHR$(&H1B);"&"
970 RETURN

```

行番号

760 コメント。

770 }
 { } Vol.12, No.2 参照。
 860 }

870 最終行を送信し、ファイルをクローズするとともにロジカルエンドレコード\$\$\$
 EOF送信する。

880 }
 { } Vol.12, No.2 参照。
 970 }

```

1000 'File Hander Routine
1010 FHR=1
1020 GOSUB 1140:COLOR 5:LOCATE 25,1:PRINT "File Hander Program":LOCATE 0,3
1030 PRINT "File Listing Program .....1"
1040 PRINT "Line no. and ' append Program.....2"
1050 PRINT "Line no. and ' Delete Program.....3"
1060 PRINT "Program end .....4"
1070 INPUT "Enter Program no.";PNUM
1080 IF PNUM=1 THEN GOSUB 1140:GOTO 1000
1090 IF PNUM=2 THEN GOSUB 1330:GOTO 1000
1100 IF PNUM=3 THEN GOSUB 1550:GOTO 1000
1110 IF PNUM=4 THEN 1130
1120 PRINT "Illegal number":GOTO 1070
1130 GOSUB 1140:FHR=0:GOTO 130
1140 PRINT CHR$(8):RETURN
1150 REC$=RIGHT$(REC$,LEN(REC$)-INSTR(REC$,CHR$(8))):RETURN

```

行番号

- 1000 コメント。今回新規に追加したルーチン。
- 1010 ファイルハンドラー処理ルーチンフラグをONにする。
- 1020 画面をクリアし、タイトルを表示する。
- 1030 }
 - 1040 } 処理メニューを表示する。
 - 1050 }
 - 1060 }
- 1070 処理メニューのプログラム番号を入力。
- 1080 プログラム番号1なら、ファイルの内容を表示するプログラムへ。
- 1090 プログラム番号2なら、ライン番号とアポストロフィー(')を付加してコピーする。
- 1100 プログラム番号3なら、ライン番号とアポストロフィー(')を削除してコピーする。
- 1110 プログラム番号4なら、EXITルーチンへ。
- 1120 プログラム番号1～4以外はメッセージを出力し、再度入力へ。
- 1130 画面をクリアし、ファイルハンドラー処理ルーチンフラグをOFFにして、ターミナルモードへ。
- 1140 画面クリアサブルーチン。
- 1150 入力データの前にある不用なLFコード(16進0A)を削除するサブルーチン。

```

1160 GOSUB 1140:COLOR 5:LOCATE 25,1:PRINT "FDD File Listing Program":LOCATE 0,3
1170 INPUT "Enter listing file name";F$
1180 IF F$="FILES" OR F$="files" THEN FILES:GOTO 1170
1190 IF F$="" THEN 1300
1200 COLOR 4:PRINT
1210 OPEN F$ FOR INPUT AS #2
1220 LINE INPUT #2,REC$
1230 GOSUB 1150
1240 IF LISTING THEN LPRINT REC$
1250 PRINT REC$
1260 IF EOF(2) THEN 1290
1270 IF LSW THEN LSW=0:GOTO 1290
1280 GOTO 1220
1290 CLOSE #2
1300 INPUT "Return to main y/n";YN$
1310 IF YN$="y" OR YN$="Y" THEN 1000
1320 GOTO 1160

```

行番号

- 1160 画面をクリアし、タイトルを表示する。
- 1170 内容を表示するファイル名を入力。
- 1180 「FILES」又は「files」が入力された時は、フロッピーディスクのディレクトリーを表示。
- 1190 復改のみの時は、EXITルーチンへ。
- 1200 カラー指定をグリーンにする。
- 1210 ファイルオープン。
- 1220 ファイルより一行分一括入力。
- 1230 有効データ前にある不用なLFコード(16進0A)を削除する。
- 1240 ハードコピー指定があれば、プリント出力。
- 1250 データを画面に表示。
- 1260 入力データが終れば、EXITルーチンへ。
- 1270 STOP キーが押下されていれば、EXITルーチンへ。
- 1280 データ入力へ戻る。
- 1290 ファイルクローズ。
- 1300 メニュー画面にもどるかどうかを入力。
- 1310 「Y」又は「y」ならメニュー画面に戻る。
- 1320 タイトル表示に戻る。


```

1330 GOSUB 1140:COLOR 5:LOCATE 25,1:PRINT "Line no. and ' Append Program":LOCATE
 0,3
1340 INPUT "Enter input file name";I$
1350 IF I$="FILES" OR I$="files" THEN FILES:GOTO 1340
1360 IF I$="" THEN 1520
1370 INPUT "Enter output file name";O$
1380 COLOR 4:PRINT
1390 OPEN I$ FOR INPUT AS #2
1400 OPEN O$ FOR OUTPUT AS #3
1410 NO=0
1420 LINE INPUT #2,REC$
1430 GOSUB 1150
1440 NO=NO+10
1450 PRINT #3,NO;" ";REC$:PRINT NO;" ";REC$
1460 IF LISTING THEN LPRINT NO;" ";REC$
1470 IF EOF(2) THEN 1500
1480 IF LSW THEN LSW=0:GOTO 1500
1490 GOTO 1420
1500 CLOSE #2,#3
1510 COLOR 4
1520 INPUT "Return to main y/n";YN$
1530 IF YN$="Y" OR YN$="y" THEN 1000
1540 GOTO 1330

```

行番号

- 1330 画面をクリアし、タイトルを表示する。
- 1340 入力するファイル名を入力。
- 1350 「FILES」又は「files」が入力された時は、フロッピーディスクのディレクトリーを表示。
- 1360 復改のみの時は、EXITルーチンへ。
- 1370 出力するファイル名を入力。
- 1380 カラー指定をグリーンにする。
- 1390 入力ファイルオープン。
- 1400 出力ファイルオープン。
- 1410 ナンバーをクリア。
- 1420 一行分一括入力。
- 1430 有効データ前にある不用なLFコード(16進0A)を削除する。
- 1440 ナンバーを10番ずつ上げる。
- 1450 ナンバーとアポストロフィー(')を付加して、データを出力する。
- 1460 ハードコピーの指定があれば、プリント出力。
- 1470 入力データがなければ、EXITルーチンへ。
- 1480 ストップ キーが押下されていれば、EXITルーチンへ。
- 1490 データ入力へ戻る。
- 1500 クローズ。
- 1510 カラー指定をグリーンにする。
- 1520 メニュー画面にもどるかどうかを入力。
- 1530 「Y」又は「y」ならメニュー画面に戻る。
- 1520 タイトル表示に戻る。

```

1550 GOSUB 1140:COLOR 5:LOCATE 25,1:PRINT "Line no. and ' delete Program":LOCATE
    0,3
1560 INPUT "Enter input file name";I$
1570 IF I$="FILES" OR I$="files" THEN FILES:GOTO 1560
1580 IF I$="" THEN 1730
1590 INPUT "Enter output file name";O$
1600 COLOR 4:PRINT
1610 OPEN I$ FOR INPUT AS #2
1620 OPEN O$ FOR OUTPUT AS #3
1630 LINE INPUT #2,REC$
1640 GOSUB 1150
1650 REC$=RIGHT$(REC$,LEN(REC$)-INSTR(REC$,CHR$(&H27)))
1660 PRINT #3,REC$:PRINT REC$
1670 IF LISTING THEN LPRINT REC$
1680 IF EOF(2) THEN 1710
1690 IF LSW THEN LSW=0:GOTO 1580
1700 GOTO 1630
1710 CLOSE #2,#3
1720 COLOR 4
1730 INPUT "Return to main y/n";YN$
1740 IF YN$="y" OR YN$="Y" THEN 1000
1750 GOTO 1550

```

行番号

- 1550 画面をクリアし、タイトルを表示する。
- 1560 入力するファイル名を入力。
- 1570 「FILES」又は「files」が入力された時は、フロッピーディスクのディレクトリーを表示。
- 1580 復改のみの時は、EXITルーチンへ。
- 1590 出力するファイル名を入力。
- 1600 カラー指定をグリーン。
- 1610 入力ファイルオープン。
- 1620 出力ファイルオープン。
- 1630 一行分一括入力。
- 1640 有効データ前にある不用なLFコード(16進0A)を削除する。
- 1650 ラインナンバーとアポストロフィー(')を削除する。
- 1660 データ出力。
- 1670 ハードコピー指定があれば、プリント出力。
- 1680 入力データがなければEXITルーチンへ。
- 1690 STOP キーが押下されていれば、EXITルーチンへ。
- 1700 データ入力へ戻る。
- 1710 クローズ。
- 1720 カラー指定をグリーンにする。
- 1730 メニュー画面にもどるかどうかを入力。
- 1740 「Y」又は「y」ならメニュー画面に戻る。
- 1750 タイトル表示に戻る。

4. 使用 方 法

使用方法については前号 (Vol.12, No.2) とほぼ同じであるが、1200 BPSではプリンターの速度からハードコピーがとれないので、ターミナルプログラムにファイルハンドラー処理ルーチンを追加し、**ESC** キーを押下することによりそのルーチンを実行する様にした。

ターミナルプログラムの使用方法是、前号 (Vol.12, No.2) を参照し、ファイルハンドラー処理ルーチンについて使用方法を説明する。

ESC キーを押下すると

File Handler Program

```
File Listing Program .....1
Line no. and ' append Program.....2
Line no. and ' Delete Program.....3
Program end .....4
Enter Program no.?
```

が表示されるので、必要なプログラム番号を入力する。

1～4以外を入力すると

```
Illegal number
Enter Program no.
```

が表示されるので再度入力する。

1を入力すると

FDD File Listing Program

```
Enter listing file name? FILE ← ファイル名を入力
```

が表示されるので、画面に表示するファイル名を入力する。この時、**f・5** キーを押下した後、ファイル名を入力すると、画面に表示されるとともにプリント出力される。

ファイルの内容がすべて表示されると

```
Return to main y/n
```

が表示されるのでメニュー画面に戻る時は「Y」又は「y」を入力し、再度リスティングプログラムを実行させる時は、「N」又は「n」を入力する。

プログラムを中断する時は、**STOP** キーを押下すればよい。

メニュー画面に戻った時、2を入力すると

Line no. and ' Append Program

```
Enter input file name? FILE ← ファイル名を入力
Enter output file name? FILE ←
```

が表示されるので、ラインNo. とアポストロフィー (') を付加したいファイル名を入力ファイル名


とし、付加したファイル名を出力ファイル名にして入力する。

この機能は、ホストのFORTRANプログラムなどを、サービス時間外に修正する時、PC 8801の持つスクリーンエディターを利用するため、メモリ内にBASICプログラムのコメントとしてロードするためのものである。

出力ファイルをPC 8801のLODEコマンドでメモリーにロードすることができるので、スクリーンエディターで修正後、SAVE コマンド(必ずアスキータイプを指定する)、ファイルを作成しておけば、次に説明するラインNo. とアポストロフィー(')を削除するプログラムで削除し、ホストに送信することもできる。

メニュー画面で3を入力すると

Line no. and ' delete Program

Enter input file name? FILE ← 
Enter output file name? FILE ←

が表示されるので、ラインNo. とアポストロフィー(')を削除したいファイル名を入力ファイル名とし削除したファイル名を出力ファイル名にして入力する。

これにより、ラインNo. とアポストロフィー(')が削除され出力ファイルにコピーされる。

5. 最 後 に

以上1200BPSターミナルプログラムの説明を行った。前号(Vol.12, No.2)のプログラムを使用されている方は、行番号の下1桁が0でない所の追加と、行番号120から行番号291の修正だけで1200BPSで動作するはずである。行番号1000以下はファイルハンドラー処理ルーチンですので不用な方は削除してもらいたい。

1200BPSに対応する様プログラム作製にあたり、吉村 幸雄先生(四国女子大学)には有意義な助言をいただいた。ここに厚く御礼申し上げます。

附 録 ACOS と無手順で接続する時の手順

ACOSと無手順で接続する場合、無手順とは言いながらある手順を踏まないと接続できない。公衆回線網の場合300BPS、1200BPSとも伝送制御方式の仕様を、表1のように設定する。

表 1

パリティチェック	偶数(Even parity)
データ長	JIS 7ビット
ストップビット	1ビット

電話でACOSを呼び出し、音響カプラーにセットし「\$\$\$CON, TSS」又はSOHコード(16進01)を送信すると、ACOSと接続されるはずである。 応答がない時は、回線スピードのセットミスの場合があるので確認し、再度送信して見る。

ターミナル・プログラムを作成しててやっかいな問題は、改行の度に送られてくるコードの問題である。

ACOSでは図1の様にテキスト以外に、CRコード(16進0D, 復改コードとも言う)、LFコード(16進0A, 改行コードとも言う)、DELコード(16進7F)が送られてくる。




図 1

これは、多くのプリンター端末では復改に時間がかかるので、ACOS ではプリンター端末が復改動作を行なう時間が十分とれるように DEL コード (16 進 7F) を送ってくる。PC 8801 では、ディスプレイに DEL コード (16 進 7F) を表示させると空白を表示するため、このまま表示すると画面への表示が途中から始まったり、場合によっては、初め 3 文字が次の行の DEL コード (16 進 7F) によって消される時がある。

これを防ぐためには DEL コード (16 進 7F) を除くか、ディスプレイに表示されない (空白も表示しない) コード変換してやればよい。

今回のプログラムでは、一括入力しているため機械語のサブルーチンで DEL コード (16 進 7F) を NULL コード (16 進 00) に変換している。

図1でもわかる様にテキストの前にCRコード(16進0D)があるため LINE INPUT 文では一括入力できない時があるので LINE INPUT 文を使用する時は、注意が必要である。

ACOS に READ コマンドを使用してデータを連続的に送信する場合は図2の様にする。




図 2